/* READ COMMENTS:

This macro applies the ADDITIVE logistic regression approach, which models the transmission of the variant allele, conditional on the parental genotypes and the offspring's QT value.
The method is described in "A method for using complete and incomplete trios to localize genes related to a quantitative trait" by Kistner and Weinberg, currently under review. “Additive” here means that the relative effect of a unit change of the QT on the conditional probability (conditional on the parental mating type) that 2 copies were inherited is assumed to be the square of the relative effect of a unit change of the QT on the conditional probability that 1 copy was inherited, where 0 copies is the referent category. In the polytomous model where heterozygous offspring serve as the referent outcome, this is equivalent to saying that the logistic model is constrained such that the QT coefficient for inheriting 0 copies is the negative of the QT coefficient for inheriting 2 copies. 
This macro assumes that the trios are all complete and that the marker under study is either di-allelic or has alleles that have been meaningfully grouped into two categories.  If some trios are incomplete, use the other macro provided in the same website. The hypothesis testing results will not depend on which of the two alleles under study is designated as the "variant".  The QT need not be normally distributed, and HWE is not required. The stratification on parental genotypes confers robustness against hidden population stratification and admixture. 

Should be used with SAS Version 8 or higher. Runs in batch. 

This code works for the following SAS data.  

Each record must consist of a M,F,C, and X, where 

M= # of copies of the variant allele that the mother carries,

F= # of copies of the variant allele that the father carries,

C= # of copies of the variant allele that the offspring carries, and

X= the quantitative trait, as measured for the offspring.

YOU MUST ENTER THE NAME OF YOUR DATA SET IN THE FIRST %LET STATEMENT BELOW!

*/

%LET YOUR_DATA=ENTERHERE;

ODS LISTING CLOSE;

DATA SAMPLE;                  /* defining mating type variables and the reference category for the model */

  SET &YOUR_DATA;

  MTYPE01=0;

  MTYPE11=0;

  MTYPE12=0;

  COUNT=C;

  POSSIBLE=1;

  IF (((M=0) AND (F=1)) OR ((M=1) AND (F=0))) THEN DO;

     MTYPE01=1;

     END;

  IF ((M=1) AND (F=1)) THEN DO;

     MTYPE11=1;

     POSSIBLE=2;

     END;

  IF (((M=1) AND (F=2)) OR ((M=2) AND (F=1))) THEN DO;

     MTYPE12=1;

     COUNT=C-1;

     END;

  IF (MTYPE01) OR (MTYPE11) OR (MTYPE12) THEN OUTPUT;

RUN;

ODS TRACE OFF;

ODS OUTPUT CONTRASTS=Ctrasts ParameterEstimates=ParmEsts;

PROC GENMOD DATA=SAMPLE ORDER=DATA;

MODEL COUNT/POSSIBLE = X MTYPE01 MTYPE11 MTYPE12 / NOINT DIST=B;

CONTRAST "X" X 1;

RUN;

ods listing;

TITLE1 "Test of Linkage and Association";

PROC PRINT DATA=Ctrasts;

RUN;

TITLE1 "Estimates of Parameters from Additive Model";

PROC PRINT DATA=PARMESTS;

RUN;

